Supporting information (model code) for:
"Hydrological drivers of record-setting water level rise on Earth's largest lake system"

```
model {
    # THROUGH DECEMBER 2014, START.STEP = 1 TO START AT JANUARY 2005
    for (j in start.step:120)   {

        ### SUPERIOR
        # OBSERVATION MODEL - CUMULATIVE STORAGE CHANGE
        y.delta_s[j] ~ dnorm(delta_s[j], j.tau.yd_s[j])
        j.tau.yd_s[j] <- tau.yd_s

        # 'TRUE' CUMULATIVE STORAGE CHANGE
        delta_s[j] <- (
            # PRECIPITATION
            sum(gamma_s[start.step:j])
            # EVAPORATION
            -sum(lambda_s[start.step:j])
            # RUNOFF
            +sum(rho_s[start.step:j])
            # CONNECTING CHANNEL OUTFLOW
            -sum(beta_s[start.step:j])
            # DIVERSION OF OGOKI LONG-LAC INTO SUPERIOR (POSITIVE)
            +sum(alpha_s[start.step:j])
        )

        ### MICHIGAN-HURON
        y.delta_m[j] ~ dnorm(delta_m[j], j.tau.yd_m[j])
        j.tau.yd_m[j] <- tau.yd_m

        delta_m[j] <- (
            sum(gamma_m[start.step:j])
            -sum(lambda_m[start.step:j])
            +sum(rho_m[start.step:j])
            -sum(beta_m[start.step:j])

            # CONNECTING CHANNEL INFLOW
            +s2m*sum(beta_s[start.step:j])
            # DIVERSION OF MICHIGAN THROUGH CHICAGO RIVER (NEGATIVE)
            -sum(alpha_m[start.step:j])
        )

        ##############################
        ## Priors
        ##############################

        ### SUPERIOR
        # PRECIPITATION WITH GAMMA PRIOR, PARAMS ARE SHAPE AND RATE
        gamma_s[j]  ~ dgamma(psi_s[m[j]], tau_s[m[j]])

        # EVAPORATION WITH NORMAL PRIOR
        lambda_s[j] ~ dnorm(mu.e_s[m[j]], prec.e_s[m[j]])

        # RUNOFF WITH LOGNORMAL PRIOR
        rho_s[j]    <- exp(log.rho_s[j])
        log.rho_s[j] ~ dnorm(lmu.r_s[m[j]], lprec.r_s[m[j]])

        # CONNECTING CHANNEL OUTFLOW WITH NORMAL PRIOR
        beta_s[j] ~ dnorm(mu.o_s[m[j]], prec.o_s[m[j]])
```

```
# DIVERSION WITH NORMAL PRIOR
alpha_s[j] ~ dnorm(mu.d_s[m[j]], prec.d_s[m[j]])

### MICHIGAN-HURON
gamma_m[j]  ~ dgamma(psi_m[m[j]], tau_m[m[j]])
lambda_m[j] ~ dnorm(mu.e_m[m[j]], prec.e_m[m[j]])
rho_m[j]    <- exp(log.rho_m[j])
log.rho_m[j] ~ dnorm(lmu.r_m[m[j]], lprec.r_m[m[j]])
beta_m[j] ~ dnorm(mu.o_m[m[j]], prec.o_m[m[j]])
alpha_m[j] ~ dnorm(mu.d_m[m[j]], prec.d_m[m[j]])

#####################################################
## LIKELIHOOD FUNCTIONS
##
## *.xn WHERE x IS THE VARIABLE
## AND n IS THE SOURCE NUMBER
##
## x:
## p = PRECIPITATION
## e = EVAPORATION
## r = RUNOFF
## o = CHANNEL OUTFLOW
##
## n:
## 1 = NOAA GLM-HMD OR
##     IN THE CASE OF OUTFLOWS
##     INTERNATIONALLY COORDINATED MEASUREMENTS
## 2 = GEM-HYDRO OR
##     IN THE CASE OF OUTFLOWS
##     INTERNATIONAL GAUGING STATIONS
#####################################################

### SUPERIOR
ys.p1[j] ~ dnorm(mus.p1[j], taus.p1)
ys.p2[j] ~ dnorm(mus.p2[j], taus.p2)
ys.e1[j] ~ dnorm(mus.e1[j], taus.e1)
ys.e2[j] ~ dnorm(mus.e2[j], taus.e2)
ys.r1[j] ~ dnorm(mus.r1[j], taus.r1)
ys.o1[j]  ~ dnorm(beta_s1[j],  taus.o1)
ys.o2[j]  ~ dnorm(beta_s2[j],  taus.o2)
ys.d[j] ~ dnorm(mus.d[j], taus.d)

mus.p1[j] <- gamma_s[j] + etas.p1[m[j]]
mus.p2[j] <- gamma_s[j] + etas.p2[m[j]]
mus.e1[j] <- lambda_s[j] + etas.e1[m[j]]
mus.e2[j] <- lambda_s[j] + etas.e2[m[j]]
mus.r1[j] <- rho_s[j] + etas.r1[m[j]]
beta_s1[j] <- beta_s[j] + etas.o1[m[j]]
beta_s2[j] <- beta_s[j] + etas.o2[m[j]]
mus.d[j] <- alpha_s[j] + etas.d[m[j]]

### MICHIGAN-HURON
ym.p1[j] ~ dnorm(mum.p1[j], taum.p1)
ym.p2[j] ~ dnorm(mum.p2[j], taum.p2)
ym.e1[j] ~ dnorm(mum.e1[j], taum.e1)
ym.e2[j] ~ dnorm(mum.e2[j], taum.e2)
ym.r1[j] ~ dnorm(mum.r1[j], taum.r1)
```

```
    ym.o1[j]  ~ dnorm(beta_m1[j],  taum.o1)
    ym.o2[j]  ~ dnorm(beta_m2[j],  taum.o2)
    ym.d[j] ~ dnorm(mum.d[j], taum.d)

    mum.p1[j] <- gamma_m[j] + etam.p1[m[j]]
    mum.p2[j] <- gamma_m[j] + etam.p2[m[j]]
    mum.e1[j] <- lambda_m[j] + etam.e1[m[j]]
    mum.e2[j] <- lambda_m[j] + etam.e2[m[j]]
    mum.r1[j] <- rho_m[j] + etam.r1[m[j]]
    beta_m1[j] <- beta_m[j] + etam.o1[m[j]]
    beta_m2[j] <- beta_m[j] + etam.o2[m[j]]
    mum.d[j] <- alpha_m[j] + etam.d[m[j]]

} # END CUMULATIVE MODEL

##############################
## "SIMULATE" LIKELIHOOD FUNCTIONS
## TO VERIFY LONG-TERM BUDGET ACCOUNTING
## AND SIMULATE COMPONENTS
##############################

for (j in 1:120){
    ### SUPERIOR
    yprime.delta_s[j] ~ dnorm(delta_s[j], tau.yd_s)
    ysprime.p1[j] ~ dnorm(mus.p1[j], taus.p1)
    ysprime.p2[j] ~ dnorm(mus.p2[j], taus.p2)
    ysprime.e1[j] ~ dnorm(mus.e1[j], taus.e1)
    ysprime.e2[j] ~ dnorm(mus.e2[j], taus.e2)
    ysprime.r1[j] ~ dnorm(mus.r1[j], taus.r1)
    ysprime.o1[j] ~ dnorm(beta_s1[j],  taus.o1)
    ysprime.o2[j] ~ dnorm(beta_s2[j],  taus.o2)
    ysprime.d[j] ~ dnorm(mus.d[j], taus.d)

    ### MICHIGAN-HURON
    yprime.delta_m[j] ~ dnorm(delta_m[j], tau.yd_m)
    ymprime.p1[j] ~ dnorm(mum.p1[j], taum.p1)
    ymprime.p2[j] ~ dnorm(mum.p2[j], taum.p2)
    ymprime.e1[j] ~ dnorm(mum.e1[j], taum.e1)
    ymprime.e2[j] ~ dnorm(mum.e2[j], taum.e2)
    ymprime.r1[j] ~ dnorm(mum.r1[j], taum.r1)
    ymprime.o1[j] ~ dnorm(beta_m1[j],  taum.o1)
    ymprime.o2[j] ~ dnorm(beta_m2[j],  taum.o2)
    ymprime.d[j] ~ dnorm(mum.d[j], taum.d)
}

#############################
## BIAS TERMS
#############################
for (i in 1:12){
    ### SUPERIOR
    etas.p1[i] ~ dnorm(0,0.01)
    etas.p2[i] ~ dnorm(0,0.01)
    etas.e1[i] ~ dnorm(0,0.01)
    etas.e2[i] ~ dnorm(0,0.01)
    etas.r1[i] ~ dnorm(0,0.01)
    etas.o1[i] ~ dnorm(0,0.01)
    etas.o2[i] ~ dnorm(0,0.01)
```

```
        etas.d[i] ~ dnorm(0,0.01)

        ### MICHIGAN-HURON
        etam.p1[i] ~ dnorm(0,0.01)
        etam.p2[i] ~ dnorm(0,0.01)
        etam.e1[i] ~ dnorm(0,0.01)
        etam.e2[i] ~ dnorm(0,0.01)
        etam.r1[i] ~ dnorm(0,0.01)
        etam.o1[i] ~ dnorm(0,0.01)
        etam.o2[i] ~ dnorm(0,0.01)
        etam.d[i] ~ dnorm(0,0.01)
    }

    ##############################
    ## PRECISION FOR OBSERVATIONS
    ##############################

    ### SUPERIOR
    tau.yd_s     ~ dgamma(0.01,0.01)
    taus.p1      ~ dgamma(0.1,0.1)
    taus.p2      ~ dgamma(0.1,0.1)
    taus.e1      ~ dgamma(0.1,0.1)
    taus.e2      ~ dgamma(0.1,0.1)
    taus.r1      ~ dgamma(0.1,0.1)
    taus.o1      ~ dgamma(0.1,0.1)
    taus.o2      ~ dgamma(0.1,0.1)
    taus.d       ~ dgamma(0.1,0.1)

    ### MICHIGAN-HURON
    tau.yd_m     ~ dgamma(0.01,0.01)
    taum.p1      ~ dgamma(0.1,0.1)
    taum.p2      ~ dgamma(0.1,0.1)
    taum.e1      ~ dgamma(0.1,0.1)
    taum.e2      ~ dgamma(0.1,0.1)
    taum.r1      ~ dgamma(0.1,0.1)
    taum.o1      ~ dgamma(0.1,0.1)
    taum.o2      ~ dgamma(0.1,0.1)
    taum.d       ~ dgamma(0.1,0.1)

    # AREAL RATIO OF LAKE SUPERIOR AREA TO LAKE MICHIGAN-HURON
    s2m <- 0.7
}
```